Playfield Graphics Tutorial

Part 4 – Playfields as Data Files

by Greg Smith

Introduction:

In Part-1 we learned how to create a 480x300-pixel playfield and move a car around.  In part two, we replaced the car with an animated lemming.  Part-3 of the Playfield Graphics Tutorial introduced the concept of Background Collision Detection.  Part-4 demonstrates how to put the collision data into a file.

This step by itself does not appear that significant, but it is a stepping stone to bigger things.  In particular, this is a step towards having multiple playfields in a single program.  A later tutorial will demonstrate how to use this feature to include multiple playfields.

The Collision Table:

Remember that the Collision Table is a two-dimensional array of characters.  It is 30x30 characters – just 900 bytes.  Each element of the array represents a 16x10-pixel region on the playfield.  If the array element has an ‘x’ in it, then the lemming is allowed to walk on that part of the playfield.  If the array element has a ‘.’ in it, the lemming is not allowed to walk there.

In Part-3 we used explicit assignment statements to set up the table.  In Part-4 we have put the name of the playfield sprite into a data file along with the collision table information.  And, we have written code to read the collision table information into memory.

The Playfield.dat file:

This new file is the “playfield.dat” file and is an ordinary text file you create and edit on the PC.  The first line of the file is the name of the playfield sprite.  Each of the remaining 30 lines is the Collision Table, just as we defined it in Part-3.

Adding the file to the application

This file is placed into the application by adding the following 2 lines into the playfield.bld file:

[files]

The Program

'''

''' playfield tutorial - part 4

''' putting the playfield in a file

'''

''' by greg smith

'''

'''

''' use c coordinates

'''

OPTION C_COORDS

'''

''' turn off escape processing for speed

'''

OPTION ESCAPE OFF

dim x as int  'position of lemming from 0-480

dim y as int  'position of lemming from 0-300

dim z_dir as int  'which direction of lemming

dim z_cnt as int  'animation image of lemming

dim sprite_field as int 'the sprite number of the playfield sprites

dim sprite_lemming as int   'the sprite number of the lemming sprites

dim lemming_left as int     'index into lemming sprite for going left

dim lemming_right as int    'index into lemming sprite for going right

dim lemming_up as int       'index into lemming sprite for going up

dim lemming_down as int     'index into lemming sprite for going down

dim collision_street as char     'the street's collision value

dim collision_table[30,31] as char 'the collision table

'''

''' update file - copy file from app to flash

'''

sub update_file(fname[] as char)

    inline File_remove(fname);

    inline _load(fname);

end sub

'''

''' initialization

'''

sub init

    update_file("playfield1.dat") 'get the .dat file into flash

    sprite_field = 0  'playfield sprite

    sprite_lemming = 31    'lemming sprite is largest possible sprite - on top

    lemming_left=0        'lemming left image

    lemming_right=5       'lemming right image

    lemming_up=10         'lemming up image

    lemming_down=15       'lemming down image

    '''

    ''' the lemming has 20 bitmaps 5 for each direction

    sprite sprite_lemming, "lemming.pic"

    '''

    ''' the lemming can be anywhere on the playfield from

    ''' x = 0-480

    ''' y = 0-300

    x=240  'center of the playfield

    y=150  'center of the playfield

    z_dir=lemming_right  ' lemming faces right

    z_cnt=0              ' animation counter

    '''

    ''' set the collision flags

    '''

    collision_street = \x\

end sub

sub repaint

    dim field_x as int   '0,1,2 for the horizontal bitmap

    dim field_y as int   '0,1,2 for the vertical bitmap

    dim field as int     '0-8 for the bitmap currently visible

    dim lemming_x as int     '0-159 for the x coord on the screen

    dim lemming_y as int     '0-99  for the y coord on the screen

    field_x = x / 160    ' compute the horizontal bitmap

    field_y = y / 100    ' compute the vertical bitmap

    field = field_y*3 + field_x ' combine them to compute the bitmap number

    lemming_x = x mod 160  ' compute the lemming's position on the screen

    lemming_y = y mod 100  ' compute the lemming's position on the screen

    move sprite_field, 0, 0, field  'display the current field

    move sprite_lemming, lemming_x, lemming_y, z_dir+z_cnt 'display the lemming 

    paper white  'transparent color for the lemming

    ink ltgrey

    redraw        'redraw all the sprites

end sub

'''

''' increment the z_cnt variable

'''

sub update_z_cnt

    z_cnt=(z_cnt+1) MOD 5

end sub

'''

''' check for collision against table

'''

function collision_check(x as int, y as int) as char

    collision_check = collision_table[y/10,x/16]

end function

'''

''' process the keyboard

'''

sub get_keys

    dim tmp as int

    if key(#KEY_UP) then   'move the lemming up

        tmp=y-3

        if collision_check(x,tmp) = collision_street then

            y=tmp

            z_dir=lemming_up

            update_z_cnt

        end if

    end if

    if key(#KEY_DOWN) then 'move the lemming down

        tmp=y+3

        if collision_check(x,tmp) = collision_street then

            y=tmp

            z_dir=lemming_down

            update_z_cnt

        end if

    end if

    if key(#KEY_LEFT) then 'move the lemming left

        tmp=x-3

        if collision_check(tmp,y) = collision_street then

            x=tmp

            z_dir=lemming_left

            update_z_cnt

        end if

    end if

    if key(#KEY_RIGHT) then 'move the lemming right

        tmp=x+3

        if collision_check(tmp,y) = collision_street then

            x=tmp

            z_dir=lemming_right

            update_z_cnt

        end if

    end if

end sub

'''

''' get a string from the file

''' a string ends in CR/LF

'''

sub getstring(f as int, s[] as char)

    dim x as char

    dim n as int

    n=0

    while(true)

        get f,,x

        if x<>13 then

            if x=10 then exit while

            s[n] = x

            s[n+1] = 0

            n=n+1

        endif

    wend

end sub

'''

''' read the playfield file defined by

''' 'field' and set up the sprites and

''' the collision table

'''

sub init_playfield(field as int)

    dim fname[32] as char

    dim s[32] as char

    cls

    print "Loading..."

    ''' create the filename to read

    sprint fname, "playfield", field, ".dat"

    ''' open the playfield definition file

    open fname for read as 1

    getstring(1, s) 'get the sprite name

    '''

    ''' the playfield sprite has 9 bitmaps

    ''' one for each of the different areas of the

    ''' playfield

    sprite sprite_field, s

    '''

    ''' read the 30 playfield collision rows

    '''

    for i=0 to 29

        getstring(1, collision_table[i])

    next

    '''

    ''' close the playfield.dat file

    '''

    close 1

end sub

'''

''' main subroutine

'''

sub mainsub

    init

    init_playfield(1)

    while true

        inline _escape(0); /* check the escape key */

        repaint

        get_keys

    wend

end sub

'''

''' call the main subroutine

'''

mainsub

The Variables

No new variables are introduced.

Option Escape Off

Normally, B2C inserts a check for the ESC key at the top of every loop.  This check takes 1/10 of a second to perform.  For simple programs, this is not a problem.  But as a program becomes more advanced, there are more and more loops and the ESC key check becomes a serious performance bottleneck.  Specifying OPTION ESCAPE OFF tells B2C not to do ESC checking.  Thus enhancing performance.   However, it is now up to the programmer to insert the following line of code into their program at the top of a major loop:

inline _escape(0);

This will check the ESC key and display a dialog box if the key is detected.

sub update_file

Update_file() will remove the specified file from the flash memory of the Cybiko and replace it with the same file from inside the application’s archive.  This is useful for files like .dl’s and data files (which must be in the flash memory to be opened by B2C).

The first command “inline File_remove(fname)” is a call to the Cybiko OS to remove the file in question from the flash.  The second command “inline _load(fname)” is a call to the B2C runtime library to copy a file from the archive into flash memory.

We will use this command to copy the “playfield1.dat” file from the application archive into flash memory.

sub init

The sprite command defining the playfield has been moved to the init_playfield() function.

sub repaint

there were no changes to this function.

sub update_z_cnt

No changes here.

function collision_check

No changes here.

sub get_keys

No changes here.

sub getstring

This subroutine reads from an opened file until a carriage-return / line-feed is detected and returns the string in a variable.  This is a very handy function which you may want to reuse in other programs.

sub init_playfield

This function has changed significantly.  We construct the filename by appending the field variable to the word “playfield” and then add on a “.dat”.  Thus, when field=1 the filename becomes “playfield1.dat”.

Then, we open a file with that name for read.  We read the first line of the file to get the sprite name of the playfield.  Finally we call the sprite command to load the sprite into memory.

Next we read the 30 lines of collision table data into the collision table.

And lastly we close the file and return.

sub mainsub

The mainsub now calls the INIT_PLAYFIELD subroutine with a parameter of ‘1’.   Additionally, it calls the “_escape(0)” function from the C library using the inline command.  This tests the keyboard for the ESC key.  We do this just once at the top of the loop because it takes about 1/10 of a second to do this check.  Checking at the top of each loop (see OPTION ESCAPE description, above) takes too much time and slows the program perceptibly.








































